Newsgroup: it.comp.lang.visual-basic
Titolo: Aggiornamento db

Ciao
Ho creato un db per un invetario.
E possibile inserire all'apertura del db una finestra dove c'è indicato la
data del file (Aggiornamento del db)?
Grazie

 E possibile inserire all'apertura del db una finestra dove
c'è indicato la data del file (Aggiornamento del db)?
Parliamo di visual basic, giusto? Allora potresti fare
qualcosa del genere:
Option Explicit
Const mc_strDBSNAME As String = "C:\WINDOWS\DESKTOP\DB1.MDB"
Private Sub Form_Load()
  Call Show
  Call Refresh
  Call MsgBox( _
      "Data ultimo aggiornamento: " _
    & Format$(FileDateTime(mc_strDBSNAME), "General Date") _
    , vbInformation _
    , mc_strDBSNAME)
End Sub
--
Ciao.
Maurizio

 ha scritto nel messaggio
news:384e6...@news.megasys.it...


 Ma questo va bene anche con visual basic di Access
Ohibo`. Immagino che la tua sia una domanda. Se e` cosi`
allora:
1. Senza malizia ne' malignita`, ma semplice osservazione o
indicazione: non facevi prima a provare?
2. Access 97 o 2000? Si`, in quanto FileDateTime e` membro
della classe FileSystem della libreria VBA che e` appunto
presente nei riferimenti sia dei progetti di Visual Basic (4,
5, 6) che di Access (97, 2000).
3. Nello specifico caso di Access conviene fare qualche
ulteriore considerazione.
3.a Invece di aprire una maschera potresti scrivere una macro
"Autoexec".
3.b. L'argomento da passare a FileDateTime, invece che
mantenerlo nel codice come costante (o altro modo eventuale),
potra` essere ottenuto direttamente dalla proprieta` Name
della proprieta` CurrentDb (che restituisce un oggetto
Database della libreria DAO) della classe Application di
Access:
    VBA.Strings.Format$( _
        VBA.FileSystem.FileDateTime( _
 
Access.Application.CurrentDb.Name
) _
      , "General Date")
ovvero, abbreviato:
    Format$(FileDateTime(CurrentDb.Name), "General Date")
3.c. Trattandosi di un database potrebbe non essere
significativa per te la data ottenuta per mezzo di
FileDateTime in quanto informa solamente riguardo l'apertura
del file e cambia anche con una semplice apertura
immediatamente seguita da una chiusura. Potrebbe interessarti
piuttosto la proprieta` LastUpdate piu` recente fra quelle
delle tabelle...
--
Ciao.
Maurizio

 ha scritto nel messaggio
news:384f3...@news.megasys.it...

Si grazie ma non mi funziona.
Non so come inserirlo nel database di access2000
Mi puoi ancora aiutare?
Sei molto gentile.
Ciao

 Non so come inserirlo nel database di access2000
In un modulo potresti scrivere questa funzione:
Option Compare Database
Option Explicit
Public Function ShowMessageLastUpdatedTable( _
)
  Dim dtmMaxDate  As Date
  Dim daoDocument As DAO.Document
  With Access.Application.CurrentDb
    With .Containers.Item("Tables")
      For Each daoDocument In .Documents
        With daoDocument
          Debug.Print _
            .Name; " "; .LastUpdated
          If dtmMaxDate < .LastUpdated _
          Then _
            Let dtmMaxDate = .LastUpdated
        End With: Set daoDocument = Nothing
      Next
    End With
  End With
  Call MsgBox( _
      "Data ultimo aggiornamento:" & vbCrLf _
    & Format$( _
          dtmMaxDate _
        , "dddd d mmmm yyyy h:nn:ss") _
    , vbInformation)
End Function
che poi chiamerai dalla macro "AutoExec" mediante l'azione
"EseguiCodice".
Verifica il codice perche' l'ho scritto alla valachevaibene.
--
Ciao.
Maurizio

 ha scritto nel messaggio
news:38508...@news.megasys.it...

Ho inserito il codice come tu dici ma esce fuori un errore:
ERRORE DI COMPILAZIONE:
IMPOSSIBILE TROVARE IL PROGETTO O LA LIBREIA.
Cos'è che manca?
Ciao e grazie.

 Cos'è che manca?
Probabilmente il riferimento a:
    Microsoft DAO v.r Library
dove v sta per numero versione e r per numero release. O
qualche altro riferimento come, per esempio, VBA (?!).
Che riferimenti ci sono nel tuo progetto?
--
Ciao.
Maurizio

 ha scritto nel messaggio
news:38520...@news.megasys.it...


 Ne manca qualcuno?
Direi che quelli che servono ci sono tutti!... Non so che
dirti. Hai qualche altro sintomo o indizio per aiutarmi ad
aiutarti?
--
Ciao.
Maurizio